
Subject: Wealth Index data disaggregation  
Posted by Nahid on Sun, 01 Jan 2023 18:09:38 GMT
View Forum Message <> Reply to Message

Would it be possible to disaggregate the wealth index (windex5) with the Rural-Urban once the
rural wealth index (windex5r) and urban wealth index (windex5u) are unavailable?     

Subject: Re: Wealth Index data disaggregation  
Posted by Bridgette-DHS on Wed, 04 Jan 2023 01:31:26 GMT
View Forum Message <> Reply to Message

Following is a response from Senior DHS staff member, Tom Pullum:

All the wealth quintiles are constructed with the HR file and the continuous index, hv271.  The
urban and rural versions are just constructed separately for the urban households and the rural
households.  The following code will match DHS exactly, except for perhaps a couple of
households that are involved in ties when the continuous values are sorted.

Note that you have to copy the constructed values (in the HR file) to the PR and other files, with
merges using just the cluster and household codes (and region for an India survey). Everyone
(woman, man, and child) in the same household has the same value of the continuous wealth
index and the quintiles.

The program is set up for a survey that already includes the urban and rural quintiles, so that you
can check the calculation. You would have to change the program slightly for a survey that
actually does NOT include the urban and rural quintiles.

* SIMPLE STATA CODE TO CONSTRUCT WEALTH QUINTILES THAT ALMOST MATCH DHS

set more off
use "...\HTHR71FL.DTA" , clear

keep hv001 hv002 hv005 hv012 hv013 hv025 hv270* hv271*

* calculate overall quintiles
gen mem = hv012
replace mem = hv013 if mem == 0
gen pwt=mem*hv005
gen wt=pwt/1000000
xtile hv270_test=hv271 [pweight=pwt], nquantiles(5)
tab hv270 hv270_test

* calculate urban quintiles
xtile hv270_test1=hv271 if hv025==1 [pweight=pwt], nquantiles(5)
tab hv270a hv270_test1 if hv025==1

* calculate rural quintiles

xtile hv270_test2=hv271 if hv025==2 [pweight=pwt], nquantiles(5)
tab hv270a hv270_test2 if hv025==2

gen hv270a_test=.
replace hv270a_test=hv270_test1 if hv025==1
replace hv270a_test=hv270_test2 if hv025==2

correlate hv270a hv270a_test

